A Wizard's Quest: Spinning Black Holes in *UD*

Brian Beckman
Feb 2026





Black Hole Parameters
Spin (a)
0.99
Orbit Parameters
Initial Radius
1.8
Ang Mom (Phi vel)
0
Vertical Kick (Theta vel)
0
Time Dilation (E)
1.
Simulation
Duration
30
Bounding Box Scale
1.5
RESET TO DEFAULTS



Black Hole Parameters
Spin (a)
0.95
Orbit Parameters
Initial Radius
6.
Ang Mom (Phi vel)
0.05
Vertical Kick (Theta vel)
0.05
Time Dilation (E)
0.94
Simulation
Duration
5000.
Bounding Box Scale
5.
RESET TO DEFAULTS

Abstract

The Kerr metric around a spinning black hole is much more difficult than the Schwarzschild. The quest to find and solve its field equations took some 50 years from Einstein's publishing of general relativity in 1915. The *dramatis personae* include Boyer, Lindquist, Kerr, Ernst, Chandrasekhar, and others who published results from 1965 to 1968. We tell the story not historically, but as a fable of challenge and victory, in a spirit of rigorous fun. Along the way, we stress-test $\mathcal{UD}$ and the Relativity Toolkit, v1.10.0. That's the novel contribution of this notebook.

Unlike static Schwarzschild, Kerr has off-diagonal terms that couple time and rotation: ***frame-dragging***. Light cones near the black hole are tipped in the spin direction, dragging orbits out of the Keplerian plane. The effect is also called *gravitomagnetism*, due to its resemblance to electrodynamical magnetism, both producing stresses perpendicular to planes of motion.

Over time, stable Kerr orbits trace out toroidal structures via nodal precession similar to that of a Lagrange top. This precession is additional to precession of the perihelion, present even in Schwarzschild (non-rotating) geometry. Kerr black holes also have *Ergospheres*, aka *static limits*, spacetime boundaries outside the event horizons. Inside an Ergosphere, spacetime itself flows faster than light. Inflowing geodesics are swept along the spinning direction of the black hole.

Introduction: A Quest

The Wizard, Chandra, reveals a quest: "find the blueprints, the keys that explain this ancient artifact I have, this wondrous machine, more wondrous than Antikythera, this interactive visualizer of orbits around a spinning black hole. Without the plans, I can't understand or explain these things that the machine does!" He shows us: *The 3D Spirograph*, wherein the orbits fly up and down, out of the plane, but never escape a donut-shaped envelope; *The Hula Hoop*, wherein the orbits become Schwarzschild—planar—when the black hole stops spinning; *The Rip Tide*, wherein particles dumped straight down into the black hole touch the Ergosphere and are swept sideways and flung away. "Worse yet," the wizard says, "I can't replicate, maintain, or repair the machine. If it breaks, it's lost forever!"

The clockwork of the visualizer is an indecipherable bloc of definitions. Our quest is to explain them. To do that, we must find the keys that lie on the road. The wizard gives us a map, a linearized far-field approximation. But the wizard warns us that a dragon named ansatz Kerr stands in the way: the full vacuum field equations are too hard to look at, let alone to solve. But, the wizard gives us a formula and provisions—new compiler functions—for a magical sword, *Ernst*, that can help us.

- **The Dragon**: We write the Einstein vacuum field equations, Ricci = 0, over an ansatz metric of four diagonal components, as we did with Schwarzschild, but this time including a fifth, coupling term, conjured intuitively, in the symmetrical t—ϕ slots. While the equations are easy for $\mathcal{UD}$ to write, they are stupendously large, with 300-800 terms each. It is not feasible to see structure amongst the enormity, to see a starting point for teasing the equations apart and solving them incrementally. (TODO: perhaps AI can take a stab at solving them; I didn't try it.)

- **The Wizard's Map**: Are we on the right road? A simplified, linearized, weak-field, far-field ansatz that looks like Newton-Kepler with frame-dragging shows that, yes, we are on the right road, the keys lie beyond the dragon. But we can't slay him, and we can't step around him because he is surrounded by an impenetrable thicket of thorny algebra.
- **The Magical Sword**: a sword name *Ernst* can help us. If we can find pieces of a jewel in the thicket blocking the road, we can slot them into the hilt of the sword and enchant the monster into letting us pass. Ernst gives us a simplified equation that the Kerr metric must satisfy. We write the equation in $\mathcal{UD}$ and recapitulate his proof.
- **The Road Beyond**: We compare our metric to the Wolfram Function Repository (WFR) and find a match.
- **The Quest Achieved**: We find the vacuum equations now tractable, symbolically and numerically, and show how to build them into the visualizer. We now can explain, in detail, The 3D Spirograph, The Hula Hoop, The Rip Tide.

Along the way, the toolkit is subjected to torture test and found worthy.

Before we set out, the wizard fully provisions us for the quest.

Basic Provisions: the Relativity Toolkit

Our backpacks need supplies for the journey: enough food, water, medical supplies, ropes and tackle, and so on, to keep us alive and happy while we search for the keys to The Wondrous Machine.

Optional: Quit for a Fresh Kernel

- *Note, if you call* `Quit` *in a notebook, evaluation may hang. If you want to "Evaluate Notebook," mark the following two cells "Evaluatable" in the Cell→Property menu, call* `Quit` *and* `FrontEndTokenExecute` *once each, then comment out the cells, or mark them unevaluatable again. Alternatively, you can quit the kernel via the Evaluation menu (last item). At that point, you can "Evaluate Notebook" in a clean environment, or evaluate cells individually.*

```
Quit[];
```

In[∘]:=
```
FrontEndTokenExecute["DeleteGeneratedCells"]
```

Download the Code

In[1]:=
```
RTbranch = "main";
RTbase = "https://raw.githubusercontent.com/rebcabin/RelativityToolkit/" <>
   RTbranch <> "/";
RTbust = "?t=" <> ToString[Round[AbsoluteTime[]]]; (*Force fresh download*)
RTrules = RTbase <> "RelativityToolkit.wl" <> RTbust;
RTtests = RTbase <> "RelativityToolkit_RegressionTests.wl" <> RTbust;
RTviz = RTbase <> "RelativityToolkit_VisualShowcase.wl" <> RTbust;
Get[RTrules];
```

» Relativity Toolkit version : 1.10.0

» Relativity Connection Symbol : Γ

Provisions for the Sword: New Compiler Functions

The Wizard Chandra gave us the pieces for The Magical Sword. We are sure we will need it, but we should not put it together too early, lest we damage it or dissipate its power.

We add several new general-purpose functions to the $\mathcal{UD}$ compiler corpus. This section sketches their functionality, and the rest of the notebook illustrates their use by example. These functions pertain to future problems in general relativity and differential geometry.

In this notebook, “matrix” is a synonym for a Wolfram 2D array. A tensor is either a matrix that satisfies coordinate-transformation rules, or a collection of symbolic $\mathcal{UD}$ expressions that also satisfy coordinate-transformation rules.

A Summoning Spell: Γ-getter

Just as **`CalculateRicciComponent`**, already in the toolkit, needs a Riemann-getter for fetching components from Riemann, so **`CalculateRiemannComponent`** needs a Γ-getter for fetching components of the Christoffel symbols. Heretofore, Γ-getters were ad-hoc, constructed on-the-fly. But now we offer a way to build them from rule-patterned ancestors: the metric and its inverse.

Given symbols and rules for the covariant metric, $g_{\mu\nu}$, and its contravariant inverse, $g^{\mu\nu}$, **`CalculateGammaComponent`** is a general way to fetch an individual component at indices $\mathcal{U}[s]$, $\mathcal{D}[m]$, $\mathcal{D}[n]$ of the Christoffel symbols. This definition and implementation mirrors its partners, **`CalculateRiemannComponent`** and **`CalculateRicciComponent`**, introduced in v1.9.0.

- *The covariant and contravariant metrics stand in a mutually matrix-inverse relationship because $g^{\mu\rho}\, g_{\rho n} = \delta^{\mu}{}_{\nu}$, the Kronecker delta, aka the Identity matrix.*

In[◦]:=
```
CalculateGammaComponent[s_, m_, n_,
   gDD_Symbol, gDDRules_,
   gInvUU_Symbol, gInvUURules_,
   coords_] :=
  (ContractAll[
       ChristoffelsFromMetric[gDD, gInvUU, s, m, n],
       coords] /.
     gDDRules /.
    gInvUURules) //
   EvaluateUDPartials;
```

The Handguard: Gradient

For Ernst’s equation (the Magical Sword), we’ll need the gradient operator in curved spacetime.

$$\nabla^{\mu} f \equiv g^{\mu\nu} f_{,\nu} \tag{1}$$

is also the workhorse of Hamilton-Jacobi Theory (whence we get the geodesic equations) e.g.

$$H = \frac{1}{2} g^{\mu\nu} p_{\mu} p_{\nu} \quad \text{, where } p_{\mu} = S_{,\mu} \text{ is the derivative of the action } S \tag{2}$$

Because **GradRaised** needs only the contravariant metric, it takes only a symbol and rules for that.

```
In[◦]:= GradRaised[func_, gInvUU_Symbol, gInvUURules_, coords_] :=
          Table[Sum[(gInvUU[𝒰[μ], 𝒰[ν]] /. gInvUURules) D[func, ν],
            {ν, coords}], {μ, coords}];
```

The Hilt: Gradient Squared

The hilt of the Magical Sword requires the squared gradient.

$$(\nabla f)^2 \equiv \nabla_{\mu} f \, \nabla^{\mu} f \equiv g^{\mu\nu} f_{,\mu} f_{,\nu} \equiv \nabla f \bullet \nabla f \tag{3}$$

Likewise, GradSquared needs only the contravariant metric, taking only a symbol and rules for that.

```
GradSquared[func_, gInvUU_Symbol, gInvUURules_, coords_] :=
  With[{grad = MakeIndexer[
       GradRaised[func, gInvUU, gInvUURules, coords], coords]},
   Simplify[
    Sum[D[func, μ] × grad[μ], {μ, coords}]]];
```

The Blade: Scalar Laplacian

The blade of the Magical Sword is the Laplacian:

$$\nabla^2 f \equiv \frac{1}{\sqrt{|g|}} \partial_{\mu}\left(\sqrt{|g|}\, g^{\mu\nu} \partial_{\nu} f\right) \tag{4}$$

This is not $(\nabla f)^2$.

In addition to the now-regulatory contravariant metric, the Laplacian needs one more parameter, the square root of the determinant of the **covariant** metric, $g_{\mu\nu}$—Emphasis: the determinant of the down-down metric. This is best computed externally and passed in.

```
In[◦]:= ScalarLaplacian[func_, gInvUU_Symbol, gInvUURules_, sqrtDetg_, coords_] :=
          With[{grad = MakeIndexer[GradRaised[func, gInvUU, gInvUURules, coords], coords]},
           Simplify[
            Sum[D[sqrtDetg * grad[μ], μ], {μ, coords}] / sqrtDetg]];
```

The Wondrous Machine: Kerr Orbit Visualizer

Before we leave on the quest to find out how it works, the Wizard Chandra shows us the whole machine and demonstrates what it does.

But there is a warning label: this ancient artifact is designed to illustrate physically plausible scenarios. Given the particular decomposition of unconstrained initial conditions and parameters, it is possible to create non-physical scenarios, specifically ones that validate the null condition that velocity squared is −1.

Dependencies, Mute, En-Bloc

These will be explained as we go along the quest's journey. For now, take them as granted, as just enough to enable the visualizer.

In[8]:=
```
coords = {t, r, θ, ϕ};
Σ = r^2 + a^2 Cos[θ]^2;
Δ = r^2 - 2 M r + a^2;
```

$$\text{kerrMetric} = \begin{pmatrix} -\left(1-\frac{2\,M\,r}{\Sigma}\right) & 0 & 0 & -\frac{2\,M\,a\,r\,\text{Sin}[\theta]^2}{\Sigma} \\ 0 & \frac{\Sigma}{\Delta} & 0 & 0 \\ 0 & 0 & \Sigma & 0 \\ -\frac{2\,M\,a\,r\,\text{Sin}[\theta]^2}{\Sigma} & 0 & 0 & \left(r^2+a^2+\frac{2\,M\,a^2\,r\,\text{Sin}[\theta]^2}{\Sigma}\right)\text{Sin}[\theta]^2 \end{pmatrix};$$

```
Clear[gDD, gUU];
kerrRules = Join[
  MatrixToUDRules[kerrMetric, gDD, 𝒟, coords],
  {gDD[__] → 0}];
kerrInverse = Simplify[Inverse[kerrMetric]];
kerrInvRules = Join[
  MatrixToUDRules[kerrInverse, gUU, 𝒰, coords],
  {gUU[__] → 0}];
kerrGamma[s_, m_, n_] :=
   CalculateGammaComponent[s, m, n, gDD, kerrRules, gUU, kerrInvRules, coords];
```

Unknown functions of proper time τ

In[17]:=
```
funcs = {t[τ], r[τ], θ[τ], ϕ[τ]};
velocities = D[funcs, τ];
accelerations = D[velocities, τ];
```

The Geodesic Equation, $x'' + \Gamma x' x' = 0$

In[20]:=
```
geodesicEqs =
   With[{velLookup = MakeIndexer[velocities, coords], (* toolkit function *)
     accLookup = MakeIndexer[accelerations, coords]},
    Table[accLookup[μ] +
       Sum[(kerrGamma[μ, ν, λ] /. Thread[coords → funcs]) velLookup[ν] × velLookup[λ],
        {ν, coords},
        {λ, coords}] == 0,
     {μ, coords}]];
```

Explicit Acceleration Rules

In[21]:=
```
explicitAccelerations = Solve[
      geodesicEqs, accelerations]〚1〛 /.
    Rule → Equal;
```

The Visualizer itself

In[22]:=
```
With[{defaultDuration = 500,
   defaultEnergy = 1.2,
   defaultVerticalKick = 0.02,
   defaultAngularMomentum = 0.05,
   defaultInitialRadius = 8.0,
   defaultSpinParameter = 0.9,
   defaultPuff = 1.5}, {
   minDuration = 10, maxDuration = 10 000,
   minPuff = 1.5, maxPuff = 20,
   minSpin = 0.0, maxSpin = 0.99,
   minRadius = 1.0, maxRadius = 20.0,
   minAngMom = -0.15, maxAngMom = .15,
   minKick = 0.0, maxKick = 0.10,
   minE = 0.8, maxE = 2.0},
  {innerLimitRadius = 2.1},
 Manipulate[Module[{sol, Mval = 1, Tmax = duration, rMin},
    initConds = {t[0] == 0, t'[0] == initE,
      r[0] == initR, r'[0] == 0,
      θ[0] == π / 2, θ'[0] == initVelTheta,
      ϕ[0] == 0, ϕ'[0] == initL};
    sol = Quiet@NDSolve[
       {explicitAccelerations /. {M → Mval, a → spinParam}, initConds} /.
        {M → Mval, a → spinParam},
       funcs,
       {τ, 0, Tmax}, Method → "StiffnessSwitching", MaxSteps → 100 000];
    rMin = Quiet@MinValue[{(r[τ] /. sol)〚1〛, 0 ≤ τ ≤ Tmax}, τ];
    Show[ParametricPlot3D[{
        r[τ] * Sin[θ[τ]] * Cos[ϕ[τ]],
        r[τ] * Sin[θ[τ]] * Sin[ϕ[τ]],
        r[τ] * Cos[θ[τ]]} /. sol,
      {τ, 0, Tmax},
      PlotStyle → {If[rMin < innerLimitRadius, Red, Cyan], Thickness[0.003]},
      PlotRange → {
        {-initR * puff, initR * puff},
        {-initR * puff, initR * puff},
```

```
      {-initR*puff, initR*puff}},
    Boxed → True, Axes → False, Background → Black, ImageSize → Large],
   (*The Black Hole Horizons*)
   Graphics3D[{
     (*Outer Event Horizon:r+ =M+Sqrt(M^2-a^2)*)
     Black, Sphere[{0, 0, 0}, Mval + Sqrt[Mval^2 - spinParam^2]],
     (*Ergosphere (approximate at equator):r=2M*)
     Opacity[.25], Orange, Sphere[{0, 0, 0}, 2*Mval]}]]],
(*Controls*)
Style["Black Hole Parameters", Bold],
{{spinParam, defaultSpinParameter, "Spin (a)"},
 minSpin, maxSpin, Appearance → "Open"},
Style["Orbit Parameters", Bold],
{{initR, defaultInitialRadius, "Initial Radius"},
 minRadius, maxRadius, Appearance → "Open"},
{{initL, defaultAngularMomentum, "Ang Mom (Phi vel)"},
 minAngMom, maxAngMom, Appearance → "Open"},
{{initVelTheta, defaultVerticalKick, "Vertical Kick (Theta vel)"},
 minKick, maxKick, Appearance → "Open"},
{{initE, defaultEnergy, "Time Dilation (E)"}, minE, maxE, Appearance → "Open"},
Style["Simulation", Bold],
{{duration, defaultDuration, "Duration"},
 minDuration, maxDuration, Appearance → "Open"},
{{puff, defaultPuff, "Bounding Box Scale"},
 minPuff, maxPuff, Appearance → "Open"},
Delimiter,
Button["RESET TO DEFAULTS",
 spinParam = defaultSpinParameter;
 initR = defaultInitialRadius;
 initL = defaultAngularMomentum;
 initVelTheta = defaultVerticalKick;
 initE = defaultEnergy;
 duration = defaultDuration;
 puff = defaultPuff;,
 ImageSize → Medium]]]
```

Out[22]=



Interesting Things to Try

The 3D Spirograph (Toroidal Precession)

In the (+) mark upper right of the Manipulate box, choose the bookmark “The 3D Spirograph.” Alternatively, set parameters as follows

- Spin: 0.95 (high spin emphasizes the effect)

- Initial Radius: 6.0
- Ang Mom: 0.05
- Vertical Kick: 0.05 (push orbit out of the equator)
- Time Dilation: 0.94 (keep a safe distance for a stable orbit)
- Duration: 3000 **(try a slow-motion animation (▹) from duration 0 and sixteen clicks on the down speed control)**
- Bounding Box Scale: 5

In this bookmarked visualization, a particle traces a chaotic-looking, donut-shaped shell around the black hole. This is the Lense-Thirring Effect at a less extreme setting. The phenomenon is nodal precession.

The Setup: The black hole is spinning rapidly ($a = 0.95$). The particle has small slight vertical kick ($v_\theta = 0.05$) out of the equatorial plane.

The Result: With a non-rotating Schwarzschild black hole, this orbit would just be a tilted ellipse that stays on a single flat plane. But here, the black hole's spin drags the entire space around with it.

As the particle orbits, the plane it is traveling on is continuously twisted by the black hole's spin. The nodes, where the orbit crosses the equator, circulate over time, causing the orbit to precess not just in the radial direction, like Mercury's perihelion precession around the Sun, but in azimuth, painting a complex torus (donut) shape instead of a simple ring. Near a spinning black hole, there is no such thing as a fixed orbital plane.

The Hula Hoop (Collapse to Schwarzschild)

In the (+) mark upper right of the Manipulate box, choose the bookmark "The Hula Hoop." Alternatively, set parameters as follows

- Spin: 0.0 (**the critical change**)
- Initial Radius: 6.0 (same)
- Ang Mom: 0.05 (same)
- Vertical Kick: 0.05 (same)
- Time Dilation: 0.94 (same)
- Duration: 5000 (change)
- Bounding Box Scale: 7 (change from The 3D Spirograph)

The donut, and the apparent chaos, disappears. Try animating spin slowly (10 clicks on the down-speed control for spin).

The Rip Tide

In the (+) mark upper right of the Manipulate box, choose the bookmark "The Rip Tide." Alternatively, set parameters as follows

- Spin: 0.99
- Initial Radius: 1.8
- Ang Mom: 0
- Vertical Kick: 0
- Time Dilation: 1.0
- Duration: 30
- Bounding Box Scale: 1.5 (default)

In this bookmarked visualization, we have a situation where Newtonian intuition breaks down.

The Setup: a test particle at radius 1.8, inside the orange Ergosphere ($r < 2.0$) but outside the black event horizon ($r \approx 1.14$).

The Drop: zero angular momentum (Ang Mom $= 0$). In a Newtonian universe (or far from the black hole), a rock droped zero angular momentum falls straight into the center of gravity, like a stone dropped from a tower.

The Deadly Fate: The red trail does not fall straight in. Instead, it's violently ripped around the black hole in a spiral before plunging in.

This illustrates the Lense-Thirring effect (frame-dragging) at its extreme.

Inside the orange shell—the Ergosphere—the spin of the black hole drags spacetime along with it so fast that space itself move at the speed of light relative to a distant observer.

The boundary of the Ergosphere is the ***static limit***. Inside this boundary, it is impossible to remain "static," at a constant longitude ϕ. To stand still here would require moving against the flow of space faster than light.

Even though the particle has zero angular momentum—it's trying to fall straight in—any *coordinate grid* the particle lives on is rotating. The effect is intrinsic to the geometry. No non-singular coordinate system (gauge choice) can get rid of it. The particle is swept downstream by the rip tide of spacetime. It is forced to co-rotate with the black hole, no matter how hard it resists.

The code turns the track red because the particle is inside the danger zone ($r < 2.1$). It shows the particle being digested—forced into rotation and eventually swallowed beyond the horizon.

The Dragon: Kerr Metric *ab-initio*

The Wizard Chandra warned us: "Do not look the dragon, ansatz Kerr, in the eye, for his gaze is madness." But being brave (foolhardy?), we go straight into the breach. The dragon ansatz Schwarzschild yielded to a frontal attack, after all. Why not ansatz Kerr?

We make a trap—an ansatz metric. The beast is stationary, sleeping unchanging in time, and axisymmetric, it looks the same from any angle around its spin. For the other sleeping dragon, ansatz Schwarzschild, a diagonal metric sufficed. But this dragon, ansatz Kerr, spins. To trap it, we must add a twist: a cross-term, $E_{\mathrm{rot}}(r, \theta)$, linking time t and azimuth ϕ. This is the mathematical embodiment of frame-dragging.

We cast our net (the ansatz metric) and bind it with the Einstein vacuum field equations: $R_{\mu\nu} = 0$.

The 𝒰𝒟 Toolkit, faithful squire, does the dirty work. It calculates the Christoffel symbols and the Riemann curvature. But when we ask for the Ricci tensor—the final binding spell—the dragon breathes fire. The resulting equations are monstrosities. A single component, R_{rr}, swells to hundreds of terms, a thicket of algebra with no obvious path to simplification.

We have found the dragon, but we cannot slay it and cannot step through the thicket around it. The equations are too dense, too coupled, and too angry to yield a solution by force. We need a better plan. We need the Wizard's Map and the Magical Sword.

Strategy for this section:

- Set up ansatz for a stationary, axisymmetric metric.
- Write ***vacuum field equations***, Ricci = 0, via 𝒰𝒟.
    - Sub-strategy: Calculate Ricci components from toolkit functions `ChristoffelsFromMetric` and `CalculateRiemannComponent`.
    - Unlike Schwarzschild, find that the Ricci equations are unwieldy: it's not at all clear how to drive Wolfram to produce a result.
- Symbolically demonstrate Newtonian gravitation and mild gravitomagnetism in the weak-field limit, when the orbiting star is far from the black hole.
- Follow a historical derivation of the full solution: the Ernst Equation.

Stationary Axisymmetric Ansatz

```
In[23]:= coords = {t, r, θ, ϕ};
```

Set up unknown functions of r and θ only, meaning of "stationary and axisymmetric," i.e., doesn't depend on time t or azimuth ϕ. Assume we can find coordinate functions r and θ where the r—θ sector is diagonal (***Boyer-Lindquist coordinates***; see References)

- $A(r, \theta)$: time potential
- $B(r, \theta)$: radial metric
- $C(r, \theta)$: latitudinal metric
- $D(r, \theta)$: azimuthal, metric
- $E(r, \theta)$: rotational t—ϕ coupling / frame-dragging, the signature component of Kerr over Schwarzschild.

Keep an eye on E_{rot}.

Pick convenient names because C and D are reserved by Wolfram and E is ambiguous

```
In[24]:= params = {Ak[r, θ], Bk[r, θ], Ck[r, θ], Dk[r, θ], Erot[r, θ]};
```

Metric as a matrix; under the (−1, 1, 1, 1) sign convention: time-coupled components are negative.

In[25]:=
$$\texttt{ansatzMetric} = \begin{pmatrix} -\texttt{Ak}[r,\theta] & 0 & 0 & -\texttt{Erot}[r,\theta] \\ 0 & \texttt{Bk}[r,\theta] & 0 & 0 \\ 0 & 0 & \texttt{Ck}[r,\theta] & 0 \\ -\texttt{Erot}[r,\theta] & 0 & 0 & \texttt{Dk}[r,\theta] \end{pmatrix};$$

Rules for the $\mathcal{UD}$ toolkit

Instead of sparse matrices, we prefer rules for the non-zero components for further processing.

In[26]:=
```
Clear[gDD, gUU];
```

In[27]:=
```
ansatzRules = MatrixToUDRules[ansatzMetric, gDD, 𝒟, coords]
```

Out[27]=

$$\{\mathrm{gDD}_{t\,t} \to -\mathrm{Ak}[r,\theta],\ \mathrm{gDD}_{t\,\phi} \to -\mathrm{Erot}[r,\theta],\ \mathrm{gDD}_{r\,r} \to \mathrm{Bk}[r,\theta],\ \mathrm{gDD}_{\theta\,\theta} \to \mathrm{Ck}[r,\theta],\ \mathrm{gDD}_{\phi\,t} \to -\mathrm{Erot}[r,\theta],\ \mathrm{gDD}_{\phi\,\phi} \to \mathrm{Dk}[r,\theta]\}$$

Symbolic inverse

In[28]:=
```
ansatzInverse = Simplify[Inverse[ansatzMetric]];
ansatzInvRules = MatrixToUDRules[ansatzInverse, gUU, 𝒰, coords]
```

Out[29]=

$$\left\{\mathrm{gUU}^{t\,t} \to -\frac{\mathrm{Dk}[r,\theta]}{\mathrm{Ak}[r,\theta]\times\mathrm{Dk}[r,\theta]+\mathrm{Erot}[r,\theta]^2},\ \mathrm{gUU}^{t\,\phi} \to -\frac{\mathrm{Erot}[r,\theta]}{\mathrm{Ak}[r,\theta]\times\mathrm{Dk}[r,\theta]+\mathrm{Erot}[r,\theta]^2},\ \mathrm{gUU}^{r\,r} \to \frac{1}{\mathrm{Bk}[r,\theta]},\ \mathrm{gUU}^{\theta\,\theta} \to \frac{1}{\mathrm{Ck}[r,\theta]},\ \mathrm{gUU}^{\phi\,t} \to -\frac{\mathrm{Erot}[r,\theta]}{\mathrm{Ak}[r,\theta]\times\mathrm{Dk}[r,\theta]+\mathrm{Erot}[r,\theta]^2},\ \mathrm{gUU}^{\phi\,\phi} \to \frac{\mathrm{Ak}[r,\theta]}{\mathrm{Ak}[r,\theta]\times\mathrm{Dk}[r,\theta]+\mathrm{Erot}[r,\theta]^2}\right\}$$

Append catch-all zero rules;

In[30]:=
```
ansatzRules = Join[ansatzRules, {gDD[__] → 0}];
ansatzInvRules = Join[ansatzInvRules, {gUU[__] → 0}];
```

Ricci via the Toolkit

Toolkit function `CalculateRicciComponent` requires a Riemann-getter function. The Riemann-getter calls toolkit function `CalculateRiemannComponent`, which requires a Γ-getter function. Define that first.

Γ-getter

In[32]:=
```
ansatzGamma[s_, m_, n_] :=
  CalculateGammaComponent[s, m, n, gDD, ansatzRules, gUU, ansatzInvRules, coords];
```

Vacuum Field Equations (Ricci = 0)

In[33]:=
```
ansatzRicci[u_, v_] :=
  CalculateRicciComponent[ (*toolkit function*)
   u, v,
   Function[{a, b, c, d}, (*Riemann-getter as anon function*)
    CalculateRiemannComponent[a, b, c, d, (*toolkit function*)
     ansatzGamma, (*Γ-getter defined above*)
     coords]],
   coords];
```

It's only useful to show a smattering of these equations, so we apply `Short` to them.

In[34]:=
```
(vacuumFieldEquations = (Table[
       {a, b} → ansatzRicci[a, b] == 0,
       {a, coords}, {b, coords}] //
      Flatten //
     Association)) //
  Short[#, 8] &
```

Out[34]//Short=
```
⟨|{t, t} →
  (Bk[r, θ]^2 ((Ak[r, θ] × Dk[r, θ] + Erot[r, θ]^2) Ak^(0,1)[r, θ] Ck^(0,1)[r, θ] + Ck[r, θ]
         (-Ak[r, θ] (Ak^(0,1)[r, θ] Dk^(0,1)[r, θ] + 2 Erot^(0,1)[r, θ]^2) +
           Dk[r, θ] (Ak^(0,1)[r, θ]^2 - 2 Ak[r, θ] Ak^(0,2)[r, θ]) +
           2 Erot[r, θ] (Ak^(0,1)[r, θ] Erot^(0,1)[r, θ] - Erot[r, θ] Ak^(0,2)[r, θ]))) +
      <<1>> - Bk[r, θ] × Ck[r, θ] (-Ck[r, θ] × Dk[r, θ] Ak^(1,0)[r, θ]^2 -
         2 Ck[r, θ] × Erot[r, θ] Ak^(1,0)[r, θ] Erot^(1,0)[r, θ] + Erot[r, θ]^2 (Ak^(0,1)[r, θ]
             Bk^(0,1)[r, θ] + Ak^(1,0)[r, θ] Ck^(1,0)[r, θ] + 2 Ck[r, θ] Ak^(2,0)[r, θ]) +
         Ak[r, θ] (Ck[r, θ] (Ak^(1,0)[r, θ] Dk^(1,0)[r, θ] + 2 Erot^(1,0)[r, θ]^2) +
           Dk[r, θ] (Ak^(0,1)[r, θ] Bk^(0,1)[r, θ] + Ak^(1,0)[r, θ]
               Ck^(1,0)[r, θ] + 2 Ck[r, θ] Ak^(2,0)[r, θ])))) /
    (4 Bk[r, θ]^2 Ck[r, θ]^2 (Ak[r, θ] × Dk[r, θ] + Erot[r, θ]^2)) ==
   0, {t,
   r} →
  True,
 {t,
   θ} →
  True,
 {t,
   ϕ} →
  (Bk[r, θ]^2 ((Ak[r, θ] × Dk[r, θ] + Erot[r, θ]^2) Ck^(0,1)[r, θ] Erot^(0,1)[r, θ] +
         Ck[r, θ] (<<1>>)) + <<1>> - <<1>>) /
    (4 Bk[r, θ]^2 Ck[r, θ]^2 (Ak[r, θ] × Dk[r, θ] + Erot[r, θ]^2)) ==
```

```
  0, {r, t} → True, ≪6≫, {θ,
  ϕ} →
 True,
{ϕ,
  t} →
            ≪1≫
 ──────────────────────────── ==
 4 ≪2≫ (≪1≫ ≪1≫ + ≪1≫)
  0, {ϕ,
  r} →
 True, {ϕ,
  θ} → True, {ϕ,
  ϕ} →
                              1
 ─────────────────────────────────────────────────────────────
 4 Bk[r, θ]^2 Ck[r, θ]^2 (Ak[r, θ] × Dk[r, θ] + Erot[r, θ]^2)
  (Bk[r, θ]^2 (-((Ak[r, θ] × Dk[r, θ] + Erot[r, θ]^2) Ck^(0,1)[r, θ] Dk^(0,1)[r, θ]) +
       Ck[r, θ] (-Ak[r, θ] Dk^(0,1)[r, θ]^2 - 2 Erot[r, θ] Dk^(0,1)[r, θ] Erot^(0,1)[r, θ] +
          2 Erot[r, θ]^2 Dk^(0,2)[r, θ] + Dk[r, θ]
           (Ak^(0,1)[r, θ] Dk^(0,1)[r, θ] + 2 Erot^(0,1)[r, θ]^2 + 2 Ak[r, θ] Dk^(0,2)[r, θ]))) -
    ≪1≫^2 ≪2≫ ≪1≫ + Bk[r, θ] × Ck[r, θ] (≪1≫)) == 0|⟩
```

The components of the Ricci tensor are large, unwieldy expressions. Though not challenging for Wolfram to write, we did not see how to get Wolfram to solve. It's not instructive to print them out, but it is instructive to illustrate how big and unstructured they are, to make the point that we should not try to solve them directly, but find another way. We won't be using the `vacuumFieldEquations` above in the following sections, which take a different approach.

Exhibit a Component, R_{rr}

It's big

In[35]:=
```
ansatzRicci[r, r] // LeafCount
```
Out[35]=
```
753
```

In[36]:=
```
ExpressionGraph[ansatzRicci[r, r]] // EdgeCount
```
Out[36]=
```
558
```

In[37]:=
```
ExpressionGraph[ansatzRicci[r, r]] // VertexCount
```
Out[37]=
```
559
```

It's complicated: not a lot of regularity.

Optional: take the semicolon off the end of this this cell if you want to see an overwhelming, large graph, which is NOT necessary to understand the rest of the story.

In[38]:=
```
ExpressionTree[ansatzRicci[r, r], "Heads",
  TreeLayout → "RadialEmbedding", ImageSize → Full];
```

The Wizard’s Roadmap: Linearized Weak-Field Gravity

We make a close escape from the dragon’s fire to study the parchment Chandra slipped into our backpack. It is a map of the “Far Lands”—the weak-field limit beyond the dragon, where gravity is gentle and the dragon’s roar is but a whisper.

Chandra’s map suggests a sanity check: if our ansatz is correct, then far down the road from the black hole ($r \to \infty$), the terrifying E_{rot} term should dwindle into something recognizable. It should look like a “dipole”—a gravitational magnet.

We linearize our ansatz net, our metric, stripping away higher-order terms. We are left with an easy equation. If we solve it, and it reveals a dipole, we know that despite the smoke and fire ahead of us, we are at least confronting the right beast blocking the right road.

Instead of solving for E_{rot} directly, let’s check the *form* of the ansatz metric in the weak-field approximation, far from the black hole, assuming only t—ϕ coupling.

Weak Metric

In[39]:=
```
Clear[h];
```

Minkowski in spherical coordinates plus a small, unknown twist, $h[r, \theta]$ in place of the full E_{rot}:

In[40]:=
$$\text{weakMetric} = \begin{pmatrix} -1 & 0 & 0 & -h[r,\theta] \\ 0 & 1 & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ -h[r,\theta] & 0 & 0 & r^2\,\text{Sin}[\theta]^2 \end{pmatrix};$$

In[41]:=
```
(weakMetricInv = Simplify[Inverse[weakMetric]]) // MatrixForm
```

Out[41]//MatrixForm=
$$\begin{pmatrix} -\frac{r^2\,\text{Sin}[\theta]^2}{h[r,\theta]^2+r^2\,\text{Sin}[\theta]^2} & 0 & 0 & -\frac{h[r,\theta]}{h[r,\theta]^2+r^2\,\text{Sin}[\theta]^2} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \frac{1}{r^2} & 0 \\ -\frac{h[r,\theta]}{h[r,\theta]^2+r^2\,\text{Sin}[\theta]^2} & 0 & 0 & \frac{1}{h[r,\theta]^2+r^2\,\text{Sin}[\theta]^2} \end{pmatrix}$$

Linearize the Metric Inverse

In[42]:=
```
weakMetricInvLinearized = weakMetricInv /. {h[r, θ]^2 → 0}
```

Out[42]=
$$\left\{\left\{-1, 0, 0, -\frac{\text{Csc}[\theta]^2\,h[r,\theta]}{r^2}\right\}, \{0, 1, 0, 0\}, \left\{0, 0, \frac{1}{r^2}, 0\right\}, \left\{-\frac{\text{Csc}[\theta]^2\,h[r,\theta]}{r^2}, 0, 0, \frac{\text{Csc}[\theta]^2}{r^2}\right\}\right\}$$

Rules for the Toolkit

As usual, we prefer rules over matrices:

In[43]:=
```
weakRules = MatrixToUDRules[weakMetric, gDD, 𝒟, coords]
```

Out[43]=

$$\{gDD_{t\,t} \to -1,\ gDD_{t\,\phi} \to -h[r,\theta],\ gDD_{r\,r} \to 1,\ gDD_{\theta\,\theta} \to r^2,\ gDD_{\phi\,t} \to -h[r,\theta],\ gDD_{\phi\,\phi} \to r^2\,\mathrm{Sin}[\theta]^2\}$$

In[44]:=
```
weakInvRules = MatrixToUDRules[weakMetricInvLinearized, gUU, 𝒰, coords]
```

Out[44]=

$$\left\{gUU^{t\,t} \to -1,\ gUU^{t\,\phi} \to -\frac{\mathrm{Csc}[\theta]^2\,h[r,\theta]}{r^2},\ gUU^{r\,r} \to 1,\ gUU^{\theta\,\theta} \to \frac{1}{r^2},\ gUU^{\phi\,t} \to -\frac{\mathrm{Csc}[\theta]^2\,h[r,\theta]}{r^2},\ gUU^{\phi\,\phi} \to \frac{\mathrm{Csc}[\theta]^2}{r^2}\right\}$$

Join with zero rules, as before:

In[45]:=
```
weakRules = Join[weakRules, {gDD[__] → 0}];
weakInvRules = Join[weakInvRules, {gUU[__] → 0}];
```

Linearized Vacuum Field Equation

Proceed as before.

Γ-Getter

In[47]:=
```
weakGamma[s_, m_, n_] :=
  CalculateGammaComponent[s, m, n, gDD, weakRules, gUU, weakInvRules, coords];
```

$R_{t\phi}$ has a differential equation for $h[r, \theta]$

In[48]:=
```
vacuumWeakEq =
 CalculateRicciComponent[t, ϕ,
   Function[{a, b, c, d}, (*Riemann Getter*)
    CalculateRiemannComponent[a, b, c, d,
     weakGamma, (*Γ-Getter*)
     coords]],
   coords] == 0
```

Out[48]=

$$-\frac{-\mathrm{Cot}[\theta]\,h^{(0,1)}[r,\theta] + h^{(0,2)}[r,\theta] + r^2\,h^{(2,0)}[r,\theta]}{2\,r^2} == 0$$

Solution

This is much more tractable. We can guess then verify that $h[r, \theta] = \frac{1}{r}\sin^2(\theta)$ solves the vacuum, weak-field equation, in place of the full E_{rot}:

In[49]:=
```
testRule = h → Function[{r, θ}, 1/r Sin[θ]^2];
vacuumWeakEq /. testRule
```

Out[50]=

$$-\frac{-\frac{2\,\mathrm{Cos}[\theta]^2}{r}+\frac{2\,\mathrm{Sin}[\theta]^2}{r}+\frac{2\,\mathrm{Cos}[\theta]^2-2\,\mathrm{Sin}[\theta]^2}{r}}{2\,r^2}==0$$

In[51]:=
```
(vacuumWeakEq /. testRule) // Simplify
```

Out[51]=

True

QED■

$h \sim \frac{1}{r}$, the metric term $g_{t\phi}$ looks like a Newtonian potential. The additional factor of $\sin^2(\theta)$ causes twist like a dipole out of the Keplerian plane due to angular momentum of the black hole. This is the "Gravito-magnetic potential."

The Magical Sword: Ernst Equation for Kerr

The map confirms we are on the right path, but the dragon, ansatz Kerr, still blocks the road. We cannot slay the dragon, but perhaps we can step around him, if only we could hack through the impenetrable thicket of algebra at his sides. We cannot hack through it with ordinary, dull blades.

The wizard gave us parts to the Magical Sword, forged in 1968 by Frederick Ernst. Ernst realized that the thicket conceals facets of a fabulous jewel, that once inserted into the hilt of the blade, will enchant the dragon into letting us pass. The separate metric components, the temporal A, the azimuthal D, and the twisting E_{rot} are not chaotic fragments. They are the facets of the final, complex jewel.

Ernst defined a potential, $\mathcal{E} = \varphi + i\,\Psi$, that enchants the beast in one symbol. The five angry Einstein vacuum field equations collapse into a single, elegant motion: $\mathrm{Re}[\mathcal{E}]\,\nabla^2\mathcal{E} = \nabla\mathcal{E}\bullet\nabla\mathcal{E}$.

Our faithful squire, the $\mathcal{UD}$ Toolkit, fetches the parts: the Blade (the Scalar Laplacian) and the Hilt (the Gradient Squared). We have only to assemble them. If we can strike true with this sword, the thicket will cough up the jewels.

We follow Ernst's papers closely, starting with positing the metric, then showing it solves a physically equivalent equation. Space does not permit a full recapitulation of his papers, but references are included below.

The gist of Ernst's papers is first to notice that A, D, and E_{rot} behave like a single complex potential, combined into a single complex function $\mathcal{E}(r,\,\theta)$:

$$\mathcal{E} = 1 - \frac{2\,M}{r + i\,a\cos(\theta)} \tag{5}$$

where a is the ***specific angular momentum*** of the black hole: angular momentum J divided by mass M.

The Einstein equations reduce to a single non-linear equation (as opposed to five, coupled, non-linear equations, as in the ab-initio section above):

$$\mathrm{Re}[\mathcal{E}]\,\nabla^2\mathcal{E} = \nabla\mathcal{E}\bullet\nabla\mathcal{E} \tag{6}$$

We now proceed to show that the Kerr metric satisfies this equation.

Convenience Forms

In[52]:=
```
Σ = r^2 + a^2 Cos[θ]^2;
Δ = r^2 - 2 M r + a^2;
```

Metric Proposal

In[54]:=

$$\texttt{kerrMetric} = \begin{pmatrix} -\left(1-\frac{2\,M\,r}{\Sigma}\right) & 0 & 0 & -\frac{2\,M\,a\,r\,\mathrm{Sin}[\theta]^2}{\Sigma} \\ 0 & \frac{\Sigma}{\Delta} & 0 & 0 \\ 0 & 0 & \Sigma & 0 \\ -\frac{2\,M\,a\,r\,\mathrm{Sin}[\theta]^2}{\Sigma} & 0 & 0 & \left(r^2+a^2+\frac{2\,M\,a^2\,r\,\mathrm{Sin}[\theta]^2}{\Sigma}\right)\mathrm{Sin}[\theta]^2 \end{pmatrix};$$

Rules

In[55]:=
```
kerrRules = Join[
  MatrixToUDRules[kerrMetric, gDD, 𝒟, coords],
  {gDD[__] → 0}];
```

In[56]:=
```
kerrInverse = Simplify[Inverse[kerrMetric]];
kerrInvRules = Join[
  MatrixToUDRules[kerrInverse, gUU, 𝒰, coords],
  {gUU[__] → 0}];
```

Gradient, ∇ℰ

In[58]:=
```
gradKerr[func_] :=
 GradRaised[func, gUU, kerrInvRules, coords]
```

Gradient Squared, ∇ℰ•∇ℰ

In[59]:=
```
gradKerrSquared[func_] :=
  GradSquared[func, gUU, kerrInvRules, coords];
```

Determinants

Needed for the covariant Laplacian

In[60]:=
```
detG = Simplify[Det[kerrMetric]];
sqrtDetG = Simplify[Sqrt[-detG]];
```

Covariant Laplacian

In[62]:=
```
laplacianKerr[func_] :=
  ScalarLaplacian[func, gUU, kerrInvRules, sqrtDetG, coords];
```

Ernst Potential

In[63]:=
$$\text{ErnstPot} = 1 - \frac{2\,M}{r + i\,a\,\text{Cos}[\theta]};$$

In[64]:=
```
physAssumptions = {r ∈ Reals, θ ∈ Reals, M > 0, a ∈ Reals};
```

Split Re/Im Parts

In[65]:=
```
realPart = ComplexExpand[Re[ErnstPot]]
```

Out[65]=
$$1 - \frac{2\,M\,r}{r^2 + a^2\,\text{Cos}[\theta]^2}$$

Wrap Operators

In[66]:=
```
lapE = FullSimplify[laplacianKerr[ErnstPot], physAssumptions]
gradSqE = FullSimplify[gradKerrSquared[ErnstPot], physAssumptions]
```

Out[66]=
$$\frac{4\,M^2}{(r + i\,a\,\text{Cos}[\theta])^4}$$

Out[67]=
$$\frac{2\,M^2\left(a^2 + 2\,r\,(-2\,M + r) + a^2\,\text{Cos}[2\,\theta]\right)}{(r - i\,a\,\text{Cos}[\theta])\,(r + i\,a\,\text{Cos}[\theta])^5}$$

Ernst Equation: $\text{Re}[\mathcal{E}]\,\nabla^2\mathcal{E} - \nabla\mathcal{E}\bullet\nabla\mathcal{E}$

In[68]:=
```
lhs = realPart * lapE;
rhs = gradSqE;
```

In[70]:=
```
residual = lhs - rhs
```

Out[70]=
$$\frac{4\,M^2\left(1 - \frac{2\,M\,r}{r^2 + a^2\,\text{Cos}[\theta]^2}\right)}{(r + i\,a\,\text{Cos}[\theta])^4} - \frac{2\,M^2\left(a^2 + 2\,r\,(-2\,M + r) + a^2\,\text{Cos}[2\,\theta]\right)}{(r - i\,a\,\text{Cos}[\theta])\,(r + i\,a\,\text{Cos}[\theta])^5}$$

In[71]:=
```
residual = Simplify[residual]
```

Out[71]=
0

QED■

The Road Beyond the Dragon: Comparing to the WFR

Now that we are past the dragon, the details of the Wizard's Map are meaningful. We travel to the Great Library (the Wolfram Function Repository) to seek our prize.

We must be careful with dialects. The Wizard Chandra speaks of "specific angular momentum" ($a = J/M$), while the Library's books speak of "total angular momentum" (J). But there is a picture of the jewel, now in the hilt of the Magical Sword. We perform the translation and overlay our jewel upon theirs.

It is a perfect match. The very jewel that enchanted the dragon into letting us pass is the secret key to the Wondrous Machine! Our quest is over, and we thought we had only begun! The visualizer is no longer a mysterious artifact. The Wondrous Machine is a solved and understood instrument, ready for us to explore the Ergosphere and beyond.

Retrieving the Kerr Metric from WFR

With parameters M and a, substituting $Ma = J$ for the angular-momentum expected by WFR:

In[72]:=
```
wfrKerr = ResourceFunction["MetricTensor"][
  {"Kerr", M, M * a}, (*Name and parameters in a list*)
  {t, r, θ, ϕ} ]
(wfrMatrix = Normal[wfrKerr["MatrixRepresentation"]] // FullSimplify) // MatrixForm
```

Out[72]=

MetricTensor[Type: Covariant Symbol: $g_{\mu\nu}$ Dimensions: 4 Signature: Indeterminate]

Out[73]//MatrixForm=

$$\begin{pmatrix} -1+\frac{2 M r}{r^2+a^2 \mathrm{Cos}[\theta]^2} & 0 & 0 & -\frac{2 a M r \mathrm{Sin}[\theta]^2}{r^2+a^2 \mathrm{Cos}[\theta]^2} \\ 0 & \frac{r^2+a^2 \mathrm{Cos}[\theta]^2}{a^2-2 M r+r^2} & 0 & 0 \\ 0 & 0 & r^2+a^2 \mathrm{Cos}[\theta]^2 & 0 \\ -\frac{2 a M r \mathrm{Sin}[\theta]^2}{r^2+a^2 \mathrm{Cos}[\theta]^2} & 0 & 0 & \mathrm{Sin}[\theta]^2 \left(a^2+r^2+\frac{2 a^2 M r \mathrm{Sin}[\theta]^2}{r^2+a^2 \mathrm{Cos}[\theta]^2}\right) \end{pmatrix}$$

Just a new name for the metric defined above

In[74]:=
```
(myMatrix = kerrMetric) // MatrixForm
```

Out[74]//MatrixForm=

$$\begin{pmatrix} -1+\frac{2 M r}{r^2+a^2 \mathrm{Cos}[\theta]^2} & 0 & 0 & -\frac{2 a M r \mathrm{Sin}[\theta]^2}{r^2+a^2 \mathrm{Cos}[\theta]^2} \\ 0 & \frac{r^2+a^2 \mathrm{Cos}[\theta]^2}{a^2-2 M r+r^2} & 0 & 0 \\ 0 & 0 & r^2+a^2 \mathrm{Cos}[\theta]^2 & 0 \\ -\frac{2 a M r \mathrm{Sin}[\theta]^2}{r^2+a^2 \mathrm{Cos}[\theta]^2} & 0 & 0 & \mathrm{Sin}[\theta]^2 \left(a^2+r^2+\frac{2 a^2 M r \mathrm{Sin}[\theta]^2}{r^2+a^2 \mathrm{Cos}[\theta]^2}\right) \end{pmatrix}$$

In[75]:=
```
myMatrix === wfrMatrix
```

Out[75]=

True

QED■

The Quest Achieved: Vacuum Solution for Kerr

Confronting ansatz Kerr head-on got us burnt and torn by the thicket. Now, with the jewel inserted into the computer, it returns a single, silent character: 0.

Above, in *Ernst Equation for Kerr*, we already define rules for the Kerr metric and its inverse, the contravariant form. To review:

In[76]:=
```
kerrRules // Short
```

Out[76]//Short=

$$\left\{gDD_{t\,t} \to -1 + \frac{2\,M\,r}{r^2 + a^2\,\mathrm{Cos}[\theta]^2},\ gDD_{t\,\phi} \to -\frac{2\,a\,M\,r\,\mathrm{Sin}[\theta]^2}{r^2 + a^2\,\mathrm{Cos}[\theta]^2},\right.$$
$$gDD_{r\,r} \to \frac{r^2 + a^2\,\mathrm{Cos}[\theta]^2}{a^2 - 2\,M\,r + r^2},\ gDD_{\theta\,\theta} \to \ll 1 \gg + \ll 1 \gg,\ gDD_{\phi\,t} \to -\frac{2\,a\,M\,r \ll 1 \gg^2}{r^2 + a^2 \ll 1 \gg},$$
$$\left. gDD_{\phi\,\phi} \to \mathrm{Sin}[\theta]^2 \left(a^2 + r^2 + \frac{2\,a^2\,M\,r\,\mathrm{Sin}[\theta]^2}{r^2 + a^2\,\mathrm{Cos}[\theta]^2}\right),\ gDD[\_\_] \to 0\right\}$$

In[77]:=
```
kerrInvRules // Short[#, 2] & // FullSimplify
```

Out[77]//Short=

$$\left\{gUU^{t\,t} \to -1 - \frac{4\,M\,r\,(a^2 + r^2)}{(a^2 + r\,(-2\,M + r))\,(a^2 + 2\,r^2 + a^2\,\mathrm{Cos}[2\,\theta])},\right.$$
$$gUU^{t\,\phi} \to -\frac{4\,a\,M\,r}{(a^2 + r\,(-2\,M + r))\,(a^2 + 2\,r^2 + a^2\,\mathrm{Cos}[2\,\theta])},\ gUU^{r\,r} \to \frac{a^2 - 2\,M\,r + r^2}{r^2 + a^2\,\mathrm{Cos}[\theta]^2},$$
$$gUU^{\theta\,\theta} \to \frac{1}{r^2 + \ll 1 \gg \ll 1 \gg},\ gUU^{\phi\,t} \to -\frac{4\,a\,M\,r}{(a^2 + r\,(-2\,M + r))\,(\ll 1 \gg)},$$
$$\left. gUU^{\phi\,\phi} \to \frac{2\,(r\,(-2\,M + r) + a^2\,\mathrm{Cos}[\theta]^2)\,\mathrm{Csc}[\theta]^2}{(a^2 + r\,(-2\,M + r))\,(a^2 + 2\,r^2 + a^2\,\mathrm{Cos}[2\,\theta])},\ gUU[\_\_] \to 0\right\}$$

Einstein’s equations for the vacuum are $\mathrm{Ricci}_{\mathrm{Kerr}} = 0$. Check them.

Γ-Getter

In[78]:=
```
kerrGamma[s_, m_, n_] :=
  CalculateGammaComponent[s, m, n, gDD, kerrRules, gUU, kerrInvRules, coords];
```

Check a non-zero component

In[79]:=
```
kerrGamma[r, t, t] // FullSimplify
```

Out[79]=

$$-\frac{M\,(a^2 + r\,(-2\,M + r))\,(-r^2 + a^2\,\mathrm{Cos}[\theta]^2)}{(r^2 + a^2\,\mathrm{Cos}[\theta]^2)^3}$$

Riemann Helper

In[80]:=
```
kerrRiemann[u_, l_, n_, r_] :=
  CalculateRiemannComponent[u, l, n, r, kerrGamma, coords] // Simplify;
```

Check All Components

If this vanishes, then Kerr is a vacuum solution.

In[81]:=
```
Table[
   Sum[kerrRiemann[c, m, c, n], {c, coords}],
   {m, coords}, {n, coords}] // FullSimplify // MatrixForm
```

Out[81]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

QED■

Epilogue

We began this quest with a mysterious artifact : a visualizer that produced beautiful, chaotic orbits but whose inner workings were a riddle .

The Wizard Chandra warned us that the machinery—the Kerr metric—was too complex to understand by force. He was right. Confronting the Dragon (the *ab-initio* vacuum equations) earned us a firestorm of algebra. But we retreated only from the blaze, not from the quest. Consulting the Wizard' s Map (the linearized limit), we were sure that the keys lay ahead. We assembled the Magical Sword (the Ernst Equation) from provisions carried by our faithful squire, the $\mathcal{UD}$ Toolkit. With that sword, we pierced the thicket and found the enchanting jewels to pacify the dragon. The terrifying complexity of the spinning black hole—the frame-dragging, the time dilation, the time-azimuth coupling—collapses into a single, elegant identity The "Rip Tide" of the Ergosphere is not a glitch in the machine, it is the consequence of a twisting spacetime, described by the geometry we verified. The Visualizer is no longer a mystery. It is a verified instrument. Plus we showed that our tools are sharp enough for whatever monster lies ahead. The Dragon is tamed. The map is drawn. The quest is complete.

References

1. Ernst, F. J., *New Formulation of the Axially Symmetric Gravitational Field Problem*. Physical Review, Vol. 167, No. 5, pp. 1175-1178 (1968). [DOI: 10.1103/physrev.167.1175]
2. Ernst, F. J., *New Formulation of the Axially Symmetric Gravitational Field Problem. II.* Physical Review, Vol. 168, No. 5, pp. 1415-1417 (1968). [DOI: 10.1103/physrev.168.1415]

3. Ernst, F. J., *New Formulation of the Axially Symmetric Gravitational Field Problem. II [Erratum].* Physical Review, Vol. 172, No. 5, p. 1850 (1968). [DOI: 10.1103/physrev.172.1850.3]
4. Kerr, R. P., *Gravitational Field of a Spinning Mass as an Example of Algebraically Special Metrics.* Physical Review Letters, Vol. 11, No. 5, pp. 237-238 (1963). [DOI: 10.1103/PhysRevLett.11.237]
5. Boyer, R. H., *Maximal Analytic Extension of the Kerr Metric*. Journal of Mathematical Physics, Vol. 8, No. 2, pp. 265-281 (1967). [DOI: 10.1063/1.1705193]
6. Newman, E. T., Couch, E., Chinnapared, K., Exton, A., Prakash, A., & Torrence, R., *Metric of a Rotating, Charged Mass*. Journal of Mathematical Physics, Vol. 6, No. 6, pp. 918-919 (1965). [DOI: 10.1063/1.1704351]
7. Hartle, J. B., *Gravity: An Introduction to Einstein's General Relativity*. Benjamin Cummings (2003)."
8. Chandrasekhar, S., *The Mathematical Theory of Black Holes*. Oxford University Press (1983). Chapter 6, Section 53.
9. Beckman, Brian*. From the steam age to quantum fields: a unified approach via UD tensorial calculus*. Wolfram Community post.
10. Beckman, Brian*. General relativity in the UD calculus*. Wolfram Community post.
11. Beckman, Brian*. Formal Differential Geometry in the UD Calculus: Part 1*. Wolfram Community post.
12. Beckman, Brian. *Covariant Derivatives and Connections in the $\mathcal{UD}$ Calculus*. Wolfram Community post.
13. Beckman, Brian. *Riemann Curvature in $\mathcal{UD}$ + Compiler POC*. Wolfram Community post (to appear).
14. Beckman, Brian. *Schwarzschild Orbits in the $\mathcal{UD}$ Calculus*. Wolfram Community post (to appear).
15. Beckman, Brian. *Schwarzschild Metric from Scratch via $\mathcal{UD}$ Compilation*. Wolfram Community post (to appear).
16. Wolfram Function Repository*. MetricTensor*. For ground truth.
17. Wolfram Function Repository*. ChristoffelSymbol*. For ground truth.
18. [MTW] Misner, Charles W.; Thorne, Kip S.; Wheeler, John Archibald*. Gravitation*. Princeton University Press, 2017.